

## M 365 Excel Class Video 08: Lookup Formulas



### Table of Contents

XLOOKUP Arguments.....	2
Data Validation For List .....	4
Exact Match Lookup .....	5
Exact Match Lookup: Lookup Product Price.....	5
Lookup Record: Lookup Employee’s Record .....	6
Lookup Column: Lookup Column of Day’s Sales and Add .....	6
Exact Match or Next Smaller Lookup .....	7
Exact Match or Next Smallest Lookup: Lookup Tax Rate .....	7
XLOOKUP Can Do Vertical or Horizontal Lookup.....	8
Exact Match or Next Bigger Lookup: Lookup Pipe Size .....	9
Wild Card Lookup: “Coca Cola” matches “Coca Cola Inc.” .....	9
Lookup First When there are Duplicates: Lookup First Date Worked.....	10
Lookup Last When there are Duplicates: Lookup Last Sale for Employee .....	10
Two-Way Lookup: Lookup Tax Based on Allowances & Weekly Pay .....	11
Two Lookup Values: Lookup Product Price for a Specified City .....	12
XLOOKUP to Lookup Cell Reference: Lookup Rang of Sales Based On Start and End Month.....	13
FILTER function to Return Multiple Items from One Lookup Value.....	14
XMATCH function .....	14
XMATCH to Compare Two Lists.....	15
INDEX function .....	15
INDEX to Random Lookup: Randomly Lookup Employee .....	15
LOOKUP function (Original function from the first version of VisiCalc).....	16
Logical Tests.....	16
LOOKUP function Rather Than XLOOKUP for Complex Tax Calculation.....	17
SWITCH function.....	18
SWITCH to Lookup Tables.....	19
Power Query Exact Match Lookup .....	20
Power Query Exact Match or Next Smallest Lookup.....	21
Method 1: Rename Column, Append,Sort,FillDown,Filter.....	21
Method 2: Custom Function (very slow calculating for data sets with a lot of data): .....	21
Data Model DAX Formula Exact Match Lookup .....	22
Data Model DAX Formula Exact Match or Next Smallest Lookup.....	22

XLOOKUP Arguments

1	2	3	4	5	6
<b>XLOOKUP(lookup_value,lookup_array,return_array,[if_not_found],[match_mode],[search_mode])</b>					
1	<b>lookup_value</b> = item used to find match in <b>lookup_array</b>				
2	<b>lookup_array</b> = yields relative position to be used to retrieve item in <b>return_array</b>				
3	<b>return_array</b> = contains items that you want to lookup or retrieve * <b>lookup_array</b> and <b>return_array</b> must be the same size.				
4	<b>[if_not_found]</b> = what to put in cell if <b>lookup_value</b> is not found in <b>lookup_array</b>				

5	<b>[match_mode] =</b>	
0 - Exact match	-1 - Exact match or next smaller item	Return the index of the Exact match, if none return #N/A
1 - Exact match or next larger item	2 - Wildcard character match	
0 - Exact match	-1 - Exact match or next smaller item	Return the index of the Exact match, if not found return the index of the next smaller item
1 - Exact match or next larger item	2 - Wildcard character match	
0 - Exact match	-1 - Exact match or next smaller item	
1 - Exact match or next larger item	2 - Wildcard character match	Return the index of the Exact match, if not found return the index of the next larger item
0 - Exact match	-1 - Exact match or next smaller item	
1 - Exact match or next larger item	2 - Wildcard character match	Return the index of a wildcard match where *, ? and ~ have special meaning

**0 - Exact Match** = Default. "Quad" matches "Quad". "Quad " does NOT match "Quad". If no match and nothing in 4th argument, #N/A.

**-1 - Exact Match or Next Smaller** = Use for items like tax rates and commission rates. Values do not have to be sorted.

**1 - Exact Match or Next Bigger** = Use for items like square footage. Values do not have to be sorted.

**2 - Wildcards** => \* = zero or more characters, ? = single character. "Quad\*" finds anything that begins with "Quad". "\*?" finds any text.

## 6 [search\_mode] =

<ul style="list-style-type: none"> <li>☺ 1 - Search first-to-last</li> <li>☺ -1 - Search last-to-first</li> <li>☺ 2 - Binary search (sorted ascending order)</li> <li>☺ -2 - Binary search (sorted descending order)</li> </ul>	Perform a search starting at the first item
<ul style="list-style-type: none"> <li>☺ 1 - Search first-to-last</li> <li>☺ -1 - Search last-to-first</li> <li>☺ 2 - Binary search (sorted ascending order)</li> <li>☺ -2 - Binary search (sorted descending order)</li> </ul>	Perform a reverse search starting at the last item
<ul style="list-style-type: none"> <li>☺ 1 - Search first-to-last</li> <li>☺ -1 - Search last-to-first</li> <li>☺ 2 - Binary search (sorted ascending order)</li> <li>☺ -2 - Binary search (sorted descending order)</li> </ul>	Perform a binary search that relies on lookup_array being sorted in ascending order.
<ul style="list-style-type: none"> <li>☺ 1 - Search first-to-last</li> <li>☺ -1 - Search last-to-first</li> <li>☺ 2 - Binary search (sorted ascending order)</li> <li>☺ -2 - Binary search (sorted descending order)</li> </ul>	Perform a binary search that relies on lookup_array being sorted in descending order.

**1 - Search First to Last** = Default. When there are duplicates, it gets **FIRST ONE!**

**-1 - Search Last To First** = When there are Duplicates, it gets **LAST ONE!**


**2 - Binary search for Smallest To Biggest Sort** = Column must be sorted. Like old VLOOKUP and MATCH Approximate Match Lookup.

**2 - Binary search for Biggest To Smallest Sort** = Column must be sorted. Like old MATCH -1 Approximate Match Lookup.


## Data Validation For List

When you have a lookup table and you are doing Exact Match Lookup, you can use the unique list from the first column of the lookup table to help add a dropdown list of values to the cell that contains the lookup value for the XLOOKUP function. Here are the steps:


1. Select cell that contains the lookup value
2. Click the Data tab, then in the Data Tools group click the Data Validation button



3. In the Data Validation dialog box, select "List" from the Allow textbox dropdown
4. Click in the Source textbox and select the the unique list from the first column of the lookup table.
5. Click OK



## Exact Match Lookup



## Exact Match Lookup: Lookup Product Price

	A	B	C	D	E	F	G	H	I
1									
2		<b>Goal: Lookup product price using "Exact match" lookup</b>							
3									
4		<b>Product</b>	<b>Price</b>		<b>Formula in C5:</b>				
5		Quad	43.95		=XLOOKUP(B5,Product[Product],Product[Price])				
6									
7		Lookup Table:							
8									
9		<b>ProductID</b>	<b>Product</b>	<b>Cost</b>	<b>Price</b>				
10		A25C-4884	Aspen	14.55	26.95				
11		C20G-4398	Carlota	11.18	27.95				
12		Q22G-9634	Quad	19.34	43.95				
13		S35G-4382	Sunshine	11.25	22.95				
14		Y30C-4942	Yanaki	12.88	29.95				


## Lookup Record: Lookup Employee's Record

	A	B	C	D	E	F	G	H
1								
2		<b>Goal:</b> Lookup employee record based on EmployeeID in cell B5.						
3								
4		<b>EmployeeID</b>		<b>First</b>	<b>Last</b>	<b>StartDate</b>	<b>CellPhone</b>	<b>Department</b>
5		4369-4774		Ty	Smithe	40862	435-398-5510	Finance
6								
7		<b>Formula in D5:</b>	=XLOOKUP(B5,B10:B13,C10:G13)					
8								
9		<b>EmployeeID</b>	<b>First</b>	<b>Last</b>	<b>StartDate</b>	<b>CellPhone</b>	<b>Department</b>	
10		4369-9084	Sioux	Chin	10/6/12	206-767-2190	Accounting	
11		4369-4774	Ty	Smithe	11/15/11	435-398-5510	Finance	
12		4369-2234	Gigi	Sy	2/3/18	206-337-0288	Accounting	
13		4369-3979	Kip	Hensel	10/6/20	206-821-4452	Maintenance	

## Lookup Column: Lookup Column of Day's Sales and Add

	A	B	C	D	E	F	G	H
1								
2		<b>Goal:</b> Lookup column and then aggregate as sum.						
3		Add total units by Day selected in cell C5.						
4								
5		<b>Select Day:</b>	Day 3		<b>Formula in C6:</b>			
6		<b>Total Units</b>	10,849		=SUM(XLOOKUP(C5,C10:F10,C11:F22))			
7								
8		<b>Units by Day:</b>						
9								
10		<b>Hour</b>	<b>Day 1</b>	<b>Day 2</b>	<b>Day 3</b>	<b>Day 4</b>		
11		6:00 AM	373	1,646	1,606	738		
12		7:00 AM	38	1,198	1,719	1,888		
13		8:00 AM	300	1,828	1,285	816		
14		9:00 AM	1,221	265	484	1,277		
15		10:00 AM	930	780	246	15		
16		11:00 AM	914	227	409	443		
17		12:00 PM	1,271	945	1,924	715		
18		1:00 PM	1,840	617	144	887		
19		2:00 PM	966	1,824	268	378		
20		3:00 PM	672	1,208	1,946	528		
21		4:00 PM	1,177	822	323	760		
22		5:00 PM	304	432	495	364		

## Exact Match or Next Smaller Lookup



## Exact Match or Next Smallest Lookup: Lookup Tax Rate

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2		<b>Goal: Lookup earnings tax rate using "Exact match or next smaller lookup"</b>											
3													
4		<b>Earnings</b>	<b>TaxRate</b>		<b>Formula in C5:</b>								
5		\$1,200	10.0%		=XLOOKUP(B5,TaxRate[Earnings],TaxRate[TaxRate],0,-1)								
6													
7		<b>Lookup Table:</b>											
8					<b>Conditional Formatting Formulas:</b>								
9		<b>Earnings</b>	<b>Range</b>	<b>TaxRate</b>									
10		\$0	\$0 - \$299.99	0.0%	FALSE	FALSE	FALSE		FALSE	FALSE	FALSE		
11		\$300	\$300 - \$599.99	2.5%	FALSE	FALSE	FALSE		FALSE	FALSE	FALSE		
12		\$600	\$600 - \$999.99	7.0%	FALSE	FALSE	FALSE		FALSE	FALSE	FALSE		
13		\$1,000	\$1,000 - \$1,999.99	10.0%	TRUE	TRUE	TRUE		TRUE	TRUE	TRUE		
14		\$2,000	\$2,000 - \$4,999.99	15.0%	FALSE	FALSE	FALSE		FALSE	FALSE	FALSE		
15		\$5,000	\$5,000 or more	20.0%	FALSE	FALSE	FALSE		FALSE	FALSE	FALSE		
16													
17										J10: =LOOKUP(\$B\$5,\$B\$10:\$B\$15)=\$B10			
18										F10: =XMATCH(\$B\$5,\$B\$10:\$B\$15,-1)=ROWS(F\$10:F10)			

## XLOOKUP Can Do Vertical or Horizontal Lookup

	A	B	C	D	E	F	G	H	I	J
1										
2	<b>Goal:</b> XLOOKUP can use vertical or horizontal lookup tables.									
3										
4	<b>Student:</b>	Gigi			<b>Student:</b>	Chantel				
5	<b>% Grade:</b>	83.5%			<b>% Grade:</b>	0.94				
6	<b>Letter Grade:</b>	B			<b>Letter Grade:</b>	A				
7										
8	<b>Formula in C6:</b>					<b>Formula in F6:</b>				
9	=XLOOKUP(C5,B14:B18,C14:C18,,,-1)					=XLOOKUP(F5,F13:J13,F14:J14,,,-1)				
10										
11	<b>Vertical Lookup Table:</b>				<b>Horizontal Lookup Table:</b>					
12										
13	<b>% Grade</b>	<b>Letter Grade</b>			<b>% Grade</b>	90.0%	80.0%	70.0%	60.0%	50.0%
14	90.0%	A			<b>Letter Grade</b>	A	B	C	D	F
15	80.0%	B								
16	70.0%	C								
17	60.0%	D								
18	50.0%	F								



## Exact Match or Next Bigger Lookup: Lookup Pipe Size

	A	B	C	D	E	F	G	H
1								
2		<b>Goal:</b> Retrieve land square footage based on pipe size.						
3								
4		Land Square Footage	23,000		<b>Formula in C5:</b>			
5		Pipe Size Required	8"		=XLOOKUP(C4,B8:B14,C8:C14,"Enter value "&B8&" or smaller",1)			
6								
7		Max Land Square Footage for Pipe Size	Drain Pipe Size (in.)		<b>Implied category:</b>			
8		238,000	16"		238,000 >= Pipe Size > 134,000			
9		134,000	12"		134,000 >= Pipe Size > 83,000			
10		83,000	10"		83,000 >= Pipe Size > 46,000			
11		46,000	8"		46,000 >= Pipe Size > 22,000			
12		22,000	6"		22,000 >= Pipe Size > 14,000			
13		14,000	5"		14,000 >= Pipe Size > 3,500			
14		3,500	3"		3,500 >= Pipe Size > 0			

## Wild Card Lookup: "Coca Cola" matches "Coca Cola Inc."

	A	B	C	D	E	F	G	H	I
1									
2		<b>Goal:</b> Lookup Coca Cola and get a match for Coca Cola Inc.							
3		Example of: Wildcard lookup (synonyms: partial text lookup, fuzzy lookup)							
4									
5		Company (partial text)	City		<b>Formula in C6:</b>				
6		Coca Cola	Atlanta		=XLOOKUP("*"&B6&"*",B11:B14,C11:C14,"Not Found",2)				
7									
8		<b>Lookup Table:</b>							
9									
10		Company (full text)	City						
11		Coca Cola Inc.	Atlanta						
12		Pepsi Cola Inc.	NY						
13		RC Cola	KC						
14		Shasta Drinks	Calistoga						

### Lookup First When there are Duplicates: Lookup First Date Worked

	A	B	C	D	E	F	G	H	
1									
2			<b>Goal:</b> Lookup first date worked on project.						
3			"Get First" when there are duplicates.						
4									
5			<b>Formula in C12:</b>						
6			=XLOOKUP("x",D12:H12,\$D\$11:\$H\$11,"Didn't work")						
7			Enter and copy to rows below.						
8									
9			<b>"x" marks when employee worked on project.</b>						
10									
11			<b>Employee</b>	<b>First Date Worked</b>	<b>10/4/21</b>	<b>10/5/21</b>	<b>10/6/21</b>	<b>10/7/21</b>	<b>10/8/21</b>
12		Chin	10/6/21			x	x	x	
13		Gigi	10/5/21		x		x		
14		Sioux	10/5/21		x	x	x		
15		Chantel	10/4/21	x	x	x	x		x
16		Billy	10/7/21				x		x

### Lookup Last When there are Duplicates: Lookup Last Sale for Employee

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2			<b>Goal:</b> Lookup last sale for each sales rep.									
3			"Get Last" when there are duplicates.									
4												
5		<b>Date</b>	<b>Sales Rep</b>	<b>Sales</b>	<b>Sales Rep</b>	<b>Last Sale</b>	<b>Formula in G6:</b>					
6		10/3/21	Sioux	\$640.56	Sioux	311.12	=XLOOKUP(F6:F7,C6:C15,D6:D15,,,-1)					
7		10/3/21	Chin	\$706.02	Chin	765.99	Enter and it spills down to rows below.					
8		10/3/21	Chin	\$589.69								
9		10/3/21	Chin	\$695.31								
10		10/3/21	Sioux	\$474.26								
11		10/3/21	Sioux	\$786.13								
12		10/3/21	Sioux	\$311.12								
13		10/3/21	Chin	\$483.25								
14		10/3/21	Chin	\$696.97								
15		10/3/21	Chin	\$765.99								

## Two-Way Lookup: Lookup Tax Based on Allowances & Weekly Pay

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2		<b>Goal:</b> Lookup federal income tax deduction based on gross pay and allowances.										
3		Perform 2-way look, with row header & column header determining intersecting lookup value.										
4												
5		<b>SINGLE Person—WEEKLY Payroll Period</b>										
6		<b>If the gross pay is:</b>		<b>And the number of withholding allowances claimed is—</b>								
7		<b>At least</b>	<b>But less than</b>	0	1	2	3	4	5	6	7	
8		<b>The amount of income tax to be withheld is—</b>										
9	Hidden rows	450	460	31	20	12	4	0	0	0	0	
10		460	470	33	22	13	5	0	0	0	0	
11		470	480	34	23	14	6	0	0	0	0	
32		680	690	66	55	44	33	22	13	0	0	
33		690	700	67	56	45	34	23	14	0	0	
34		700	710	68	57	45	34	23	14	0	0	
35		710	720	69	58	46	35	23	15	0	0	
36												
37		<b>Gross Pay</b>		705								
38		<b>Allowance</b>		2								
39		<b>Fed Tax Deduction</b>		45								
40												
41		<b>Formula in D39:</b>										
42		=XLOOKUP(D37,B9:B35,XLOOKUP(D38,D7:K7,D9:K35),"Not Found",-1)										
43												
44												
45												
46		<b>Outer XLOOKUP</b> to lookup "700-710" row in "2" column.					<b>Inner XLOOKUP</b> to lookup all rows in "2" column. Placed in <b>return_array</b> of Outer XLOOKUP					
47												
48												
49												
50												
51												
52		45										
53												
54		<b>Formula in B52:</b>										
55		=XLOOKUP(D38,D7:K7,XLOOKUP(D37,B9:B35,D9:K35,"Not Found",-1))										

## Two Lookup Values: Lookup Product Price for a Specified City

	A	B	C	D	E	F
1						
2		<b>Goal:</b> Lookup price based on product & store.				
3		Two value lookup to determine price.				
4						
5		<b>Store</b>	Seattle			
6		<b>Product</b>	Quad			
7		<b>Price</b>	39.95			
8						
9		<b>Formula in C7:</b>				
10		=XLOOKUP(C5&C6,B13:B21&C13:C21,D13:D21)				
11						
12		<b>Store</b>	<b>Product</b>	<b>Price</b>		
13		Oakland	Quad	43.95		
14		Oakland	Aspen	27.95		
15		Oakland	Carlota	25.95		
16		Seattle	Quad	39.95		
17		Seattle	Aspen	24.95		
18		Seattle	Carlota	26.95		
19		Tacoma	Quad	30.95		
20		Tacoma	Aspen	27.95		
21		Tacoma	Carlota	25.95		

## XLOOKUP to Lookup Cell Reference: Lookup Rang of Sales Based On Start and End Month

	A	B	C	D	E	F	G	H	I	J	K	
1												
2		<b>Goal:</b> Allow user to select state and start/end month to then add the number of reported covid-19 cases.										
3												
4		Month →	Mar	Apr	May	Jun	Jul	Aug	Sep			
5		State ↓	Number Reported Covid-19 Cases									
6		AZ	1,287	6,359	12,288	59,279	94,795	27,825	16,672			
7		CA	7,429	41,435	61,666	118,149	264,856	210,497	106,540			
8		NV	1,112	3,885	3,595	9,863	29,632	21,140	10,752			
9		OR	687	1,820	1,733	4,413	9,836	8,221	6,796			
10		WA	6,931	8,670	7,259	13,564	23,508	14,370	12,722			
11												
12		State:	NV									
13		Start Month:	May									
14		End Month:	Aug									
15		Total Cases:	64,230									
16												
17		<b>Formula in C15:</b>										
18		=SUM(XLOOKUP(C12,B6:B10,XLOOKUP(C13,C4:I4,C6:I6):XLOOKUP(C14,C4:I4,C10:I10)))										
19												
20												
21												
22		Calc. steps: =SUM(XLOOKUP(C12,B6:B10,XLOOKUP(C13,C4:I4,C6:I6):XLOOKUP(C14,C4:I4,C10:I10)))										
23		<b>1.</b> The colon operator forces the bookend XLOOKUP function to lookup cell references:										
24		=SUM(XLOOKUP(C12,B6:B10,XLOOKUP(C13,C4:I4,C6:I6) : XLOOKUP(C14,C4:I4,C10:I10)))										
25		<b>2.</b> The two XLOOKUP functions deliver the two cell reference, E6 and H10 to create the range E6:H10:										
26		=SUM(XLOOKUP(C12,B6:B10,E6:H10))										
27		<b>3.</b> From the range, E6:H10, XLOOKUP looks up the row for "NV" and delivers the fours values:										
28		=SUM({3595,9863,29632,21140})										
29		<b>4.</b> SUM adds the result to get: <b>64,230.</b>										

## FILTER function to Return Multiple Items from One Lookup Value

A	B	C	D	E	F	G	H	I	J	K	L	M
1												
2	<b>Goal:</b> Extract student class records.											
3	Famous Excel task: One lookup value, return multiple items/records.											
4												
5	<b>Student</b>	<b>Class</b>	<b>Department</b>	<b>Grade</b>		<b>Student</b>						
6	Sioux Chin	Acc 121	Accounting	1.7		Gigi Dmitri						
7	Chantel Mimms	Busn 216	Business	3.1								
8	Dylan Franks	Busn 101	Business	2.1								
9	Sioux Chin	Acc 201	Accounting	4								
10	Chantel Mimms	Busn 218	Business	3.8								
11	Dylan Franks	Eng 201	English	2								
12	Sioux Chin	Econ 201	Economics	2.6								
13	Chantel Mimms	Busn 210	Business	3.3								
14	Dylan Franks	Busn 216	Business	1.9								
15	Gigi Dmitri	Busn 210	Business	2.6								
16	Chantel Mimms	Eng 201	English	3.1								
17	Gigi Dmitri	Econ 202	Economics	2								
18												
19												
20												
21												
22												
23												

Class	Department	Grade	Formula in G9:
Busn 210	Business	2.6	=FILTER(C6:E17,B6:B17=G6)
Econ 202	Economics	2	

Class	Department	Grade
Busn 210	Business	2.6
Econ 202	Economics	2

Formula 'before we had FILTER' in G15:

```
=IF(ROWS(G$15:G15)>COUNTIFS($B$6:$B$17,$G$6),"",INDEX(C$6:C$17,AGGREGATE(15,6,(ROW($B$6:$B$17)-ROW($B$6)+1)/($B$6:$B$17=$G$6),ROWS(G$15:G15))))
```

## XMATCH function

1	2	3	4
<code>XMATCH(lookup_value,lookup_array,[match_mode],[search_mode])</code>			
1	lookup_value = item used to find match in lookup_array		
2	lookup_array = item to be matched with lookup_value to yield a relative position. Can be a column or a row.		
3	[match_mode] = same as XLOOKUP		
4	[search_mode] = same as XLOOKUP		

lookup_value argument = item to lookup	lookup_array argument = search for lookup_value to yield a relative position. Can be a column or a row (not a table)	[match_mode] and [search_mode] arguments = same as XLOOKUP
----------------------------------------	----------------------------------------------------------------------------------------------------------------------	------------------------------------------------------------


  

`XMATCH(lookup_value,lookup_array,[match_mode],[search_mode])`

## XMATCH to Compare Two Lists

	A	B	C	D	E	F	G	H	I	J
1										
2		<b>Goal:</b> Conditionally format to names from Sunday list that are also in Saturday list.								
3										
4										
5		Saturday List	Sunday List							
6										
7		<b>Signups Sat. Event</b>	<b>Signups Sun. Event</b>	<b>In Both?</b>	<b>Formula in E8:</b>					
8		Ty	Miki	FALSE	=ISNUMBER(XMATCH(D8:D12,B8:B11))					
9		Gigi	Gigi	TRUE						
10		Sioux	Kip	FALSE	<b>Conditional Format Formula used in D8:</b>					
11		Chantel	Chantel	TRUE	=XMATCH(D8,\$B\$8:\$B\$11)					
12			Fran	FALSE						

## INDEX function



## INDEX to Random Lookup: Randomly Lookup Employee


	A	B	C	D	E	F	G	H	I	
2		<b>Goal:</b> Select random employee name (so they can win a prize).							Rows	5
3										
5		<b>Office Employees</b>	<b>Random Draw</b>					<b>Random Draw Column</b>		
6		Constance Bryant	Constance Bryant					Daryl West		
7		Santiago Lloyd						Elsie Manning		
8		Daryl West						Daryl West		
9		Elsie Manning						Lee Hale		
10		Richard Woods						Lee Hale		
11		Clay Cruz								
12		Irma Hubbard								
13		Kelly Christensen								
14		Lee Hale								
15		Maxine Cook								

**Formula in D6:**  
=INDEX(B6:B15,RANDBETWEEN(1,10))

\* F9 Key re- calculates formulas.

**Formula in I6:**  
=INDEX(B6:B15,RANDARRAY(I2,,1,ROWS(B6:B15),1))

## LOOKUP function (Original function from the first version of VisiCalc)



## Logical Tests


	A	B	C	D	E	F	G	H	I	J
1										
2		<b>Goal:</b> Use LOOKUP to have lookup_vector oriented vertically & [result_array] oriented horizontally.								
3		<b>Advantage:</b> XLOOKUP can't do this.								
4										
5		<b>Units</b>	27		<b>Formulas:</b>					
6		<b>Discounts</b>	0.15		In C6:=LOOKUP(C5,B10:B13,F9:I9)					
7		<b>Discounts</b>	#VALUE!		In C7:=XLOOKUP(C5,B10:B13,F9:I9,, -1)					
8										
9		<b>Units</b>		<b>Discounts</b>	0.0%	15.0%	25.0%	50.0%		
10		0								
11		25								
12		45								
13		75								



## LOOKUP function Rather Than XLOOKUP for Complex Tax Calculation

	A	B	C	D	E	F	G
2		<b>Goal:</b> "Approximate match" lookup when 1st column sorted & we want item from last column.					
3		<b>Advantage:</b> Enter fewer arguments, as compared to XLOOKUP. Faster to create formula with LOOKUP.					
4							
5		<b>Taxable Pay Lower Limit</b>	<b>Taxable Pay Upper Limit</b>	<b>Tax Rate</b>	<b>Upper Limit Previous Category</b>	<b>Cumulative Tax From Previous Categories</b>	<b>Tax Rule in Full:</b>
6		\$0.00	\$222	0%	\$0	\$0.00	
7		\$222.01	\$588	10%	\$222	\$0.00	\$0.00 +(TP - \$222) *10%
8		\$588.01	\$1,711	12%	\$588	\$36.60	\$36.60 +(TP - \$588) *12%
9		\$1,711.01	\$3,395	22%	\$1,711	\$171.36	\$171.36 +(TP - \$1,711) *22%
10		\$3,395.01	\$6,280	24%	\$3,395	\$541.84	\$541.84 +(TP - \$3,395) *24%
11		\$6,280.01	\$7,914	32%	\$6,280	\$1,234.24	\$1,234.24 +(TP - \$6,280) *32%
12		\$7,914.01	\$11,761	35%	\$7,914	\$1,757.12	\$1,757.12 +(TP - \$7,914) *35%
13		\$11,761.01	more	37%	\$11,761	\$3,103.57	\$3,103.57 +(TP - \$11,761) *37%
14							
15		If we make calculation manually: $\$541.84 + (\$3,690.80 - \$3,395) * 24\% = \$612.83$					
16							
17		<b>Gross Pay</b>				\$4,010.00	
18		<b>Withholding Allowance</b>				\$79.80	
19		<b># of Withholding Allowances</b>				4	
20		<b>Taxable Pay (TP). This is lookup_value:</b>				\$3,690.80	
21		<b>Federal Income Tax Withholdings?</b>				\$612.83	
22		<b>Federal Income Tax Withholdings?</b>				\$612.83	
23							
24		Formula in F21:=LOOKUP(F20,B6:F13)+(F20-LOOKUP(F20,B6:E13))*LOOKUP(F20,B6:D13)					
25		Formula in F22:=XLOOKUP(F20,B6:B13,F6:F13,,-1)+(F20-XLOOKUP(F20,B6:B13,E6:E13,,-1))*XLOOKUP(F20,B6:B13,D6:D13,,-1)					

## SWITCH function



## SWITCH to Lookup Tables

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	
2	<b>Ex</b>	Goal: Lookup correct lookup table based on product name.														
3	<b>#24</b>	Use lookup table to perform "approximate match" lookup.										<b>Situation #1:</b>		<b>Situation #2:</b>		
4												Multiple lookup tables.		One Lookup Table.		
5		<b>Units Transaction Table:</b>										Use SWITCH.		Use FILTER.		
7		<b>Product</b>	<b>Price</b>	<b>Units Sold</b>	<b>Price Discount</b>	<b>Price Discount</b>	<b>Bellen</b>	<b>Units</b>	<b>P. Discount</b>	<b>Product</b>	<b>Units</b>	<b>Price Discount</b>	<b>Product</b>	<b>Units</b>	<b>Price Discount</b>	
8		Bellen	26.95	24	0.25	0.25		0	0%	Bellen	0	0%	Bellen	0	0%	
9		Aspen	28.95	60	0.5	0.5		15	25%	Bellen	15	25%	Bellen	15	25%	
10		Bellen	26.95	50	0.4	0.4		45	40%	Bellen	45	40%	Bellen	45	40%	
11		Quad	43.95	5	0	0				Quad	0	0%	Quad	0	0%	
12		Quad	43.95	25	0.2	0.2	<b>Quad</b>	<b>Units</b>	<b>P. Discount</b>	<b>Quad</b>	20	20%	<b>Quad</b>	20	20%	
13		Bellen	26.95	96	0.4	0.4		0	0%	<b>Quad</b>	60	45%	<b>Quad</b>	60	45%	
14		Yanaki	30.95	8	0	0		20	20%	<b>Default</b>	0	0%	<b>Default</b>	0	0%	
15		Quad	43.95	124	0.45	0.45		60	45%	<b>Default</b>	15	30%	<b>Default</b>	15	30%	
16		Bellen	26.95	55	0.4	0.4				<b>Default</b>	25	50%	<b>Default</b>	25	50%	
17		Bellen	26.95	2	0	0	<b>Default</b>	<b>Units</b>	<b>P. Discount</b>							
18		Quad	43.95	5	0	0		0	0%							
19		Yanaki	30.95	19	0.3	0.3		15	30%							
20								25	50%							
21																
22		Formula in E8:=LOOKUP(D8,SWITCH(B8,\$H\$7,\$J\$8:\$K\$10,\$H\$12,\$J\$13:\$K\$15,\$J\$18:\$K\$20))														
23		Formula in F8:=LOOKUP(D8,FILTER(\$N\$8:\$O\$13,\$M\$8:\$M\$13=B8,\$N\$14:\$O\$16))														
24		Formulas entered, and manually copied to the rows below.														

# Power Query Exact Match Lookup

The screenshot shows the Power Query Editor interface with the 'Merge' dialog box open. The ribbon at the top includes 'Merge Queries', 'Append Queries', and 'Combine Files'. A black arrow points to the 'Merge Queries' button. The dialog box shows the following configuration:

- 04-Lookup-Solution** (Table 1):
 

UnitsSold	Product	PriceTF
174	Sunshine	22.95
212	Yanaki	29.95
70	Carlota	27.95
242	Aspen	26.95
29	Aspen	26.95
- 02-dProductPQ** (Table 2):
 

ProductID	Product	Cost	Price
A25C-4884	Aspen	14.55	26.95
C20G-4398	Carlota	11.18	27.95
Q22G-9634	Quad	19.34	43.95
S35G-4382	Sunshine	11.25	22.95
Y30C-4942	Yanaki	12.88	29.95
- Join Kind:** Left Outer (all from first, matching from second)
- Use fuzzy matching to perform the merge
- Fuzzy matching options:** (Expanded)

The 'Query Settings' pane on the right shows the 'APPLIED STEPS' list:

- Source
- Changed Type
- Merged Queries** (Selected)
- Expanded 02-dProductPQ
- Renamed Columns
- Appended Query
- Sorted Rows
- Filled Down
- Filtered Rows
- Added Custom

## Power Query Exact Match or Next Smallest Lookup


### Method 1: Rename Column, Append, Sort, Fill Down, Filter

- 1) Rename Units columns so that both columns have the same name.
- 2) Append discount table under fact table.
- 3) Sort Units column smallest to biggest (A to Z, Ascending), the sort Discount column biggest to smallest (Z to A, Descending).
- 4) Use Fill Down feature on Discount column.
- 5) For the Product column, filter out "null" values.

### Method 2: Custom Function (very slow calculating for data sets with a lot of data):

```
Table.AddColumn(#"Filtered Rows", "DiscountCustomFunction", each List.Max(Table.SelectRows  
    (#"03-disDiscountPQ", (IT)=>(IT)[Units]<=[Units])[Discount]))
```

## Data Model DAX Formula Exact Match Lookup



	[Price]		$f_x$ =RELATED('06-dProductDM'[Price])			
	UnitsSold	Pro...	PriceTF	Price	Discount	Add Column
1	174	Sunshine	22.95	22.95	0.375	
2	212	Yanaki	29.95	29.95	0.5	
3	70	Carlota	27.95	27.95	0.25	
4	242	Aspen	26.95	26.95	0.5	
5	29	Aspen	26.95	26.95	0	
6	64	Aspen	26.95	26.95	0.25	

## Data Model DAX Formula Exact Match or Next Smallest Lookup

$f_x$	=CALCULATE( MAX('07-disDiscountDM'[Discount]), FILTER('07-disDiscountDM','07-disDiscountDM'[Units]<='05-fSalesDM'[UnitsSold]) )					
	PriceTF	Price	Discount	Add Column		
e	22.95	22.95	0.375			
	29.95	29.95	0.5			